مكتبة

”خُذْ وأعطِي“

الإلكترونية

جامعة آل البيت " كلية الإقتصاد "

مجموعة طلابية تسعى لتوفير كل ما يلزم طلاب
كلية إدارة المال والاعمال من مواد وشروحات واسئلة بصورة الكترونية



Principles of Finance with Excel, 2nd edition

Instructor materials

Chapter 2
Time Value of Money



This chapter

- Future value
- Present value
- Net present value
- Internal rate of return
- Pension and savings plans

- Excel functions: **FV, PV, NPV, IRR, PMT, NPER**

Future value

- At interest rate *r%*, how much does a deposit today of $100 grow in *N* years?

$$100 * (1 + r)^N$$

Why?

- Suppose $r = 6\%$
- Suppose you deposit $100 in bank today
- In one year: $100*(1.06) = $106
- In two years: $106*(1.06) = $112.36
- Note that $106*(1.06)=$100*(1.06)^2
- Etc.

	A	B	C	D	E	F	G
1	**THE FUTURE VALUE OF A SINGLE $100 DEPOSIT**						
2	Initial deposit	100					
3	Interest rate	11%					
4							
5	**Year**	**Future value**					
6	0	100.00	<-- =$B$2*(1+$B$3)^A6				
7	1	111.00	<-- =$B$2*(1+$B$3)^A7				
8	2	123.21	<-- =$B$2*(1+$B$3)^A8				
9	3	136.76	<-- =$B$2*(1+$B$3)^A9				
10	4	151.81	<-- =$B$2*(1+$B$3)^A10				
11	5	168.51					
12	6	187.04					
13	7	207.62					
14	8	230.45					
15	9	255.80					
16	10	283.94					
17	11	315.18					
18	12	349.85					
19	13	388.33					
20	14	431.04					
21	15	478.46					
22	16	531.09					
23	17	589.51					
24	18	654.36					
25	19	726.33					
26	20	806.23					



Future value at different interest rates



Some terminology: Beginning vs end of year



Accumulating money

- Deposit $100 today and at the beginning of years 1, 2, …, 10
- Interest paid: 6% per year on outstanding balances
- How much will you have at the end of 10 years?

	A	B	C	D	E	F
1	**FUTURE VALUE WITH ANNUAL DEPOSITS at beginning of year**					
2	Interest	6%				
3	=E5					=(C6+B6)*$B$2
4	**Year**	**Account balance, beg. year**	**Deposit at beginning of year**	**Interest earned during year**	**Total in account at end of year**	
5	1	0.00	100.00	6.00	106.00	<-- =B5+C5+D5
6	2	106.00	100.00	12.36	218.36	<-- =B6+C6+D6
7	3	218.36	100.00	19.10	337.46	
8	4	337.46	100.00	26.25	463.71	
9	5	463.71	100.00	33.82	597.53	
10	6	597.53	100.00	41.85	739.38	
11	7	739.38	100.00	50.36	889.75	
12	8	889.75	100.00	59.38	1,049.13	
13	9	1,049.13	100.00	68.95	1,218.08	
14	10	1,218.08	100.00	79.08	1,397.16	
15						
16		Future value using Excel's FV function	$1,397.16	<-- =FV(B2,A14,-100,,1)		

Beginning vs end of period

Deposits at Beginning of Year



Total 1397.16 <-- Sum of the above

Deposits at End of Year



Total 1318.08 <-- Sum of the above

Present value

- Future value: If you deposit today, how much will you have in the future?
- **Present value**: If you are promised money in the future, how much is it worth today

- **Present value** and **Future value** are mirror images!

Present Value vs Future Value



$100 in three years is worth $83.96 today if the interest rate = 6%

$83.96 today is worth $100 in 3 years if the interest rate is 6%

Present value ↓ as interest ↑

$$83.96 = \underbrace{\frac{100}{(1.06)^3}}_{\text{present value of \$100 in 3 years at 6\%}} > \underbrace{\frac{100}{(1.35)^3}}_{\text{present value of \$100 in 3 years at 35\%}} = 40.64$$

Present value of multiple future payments

	A	B	C	D
1	**PRESENT VALUE OF AN ANNUITY: FIVE ANNUAL PAYMENTS OF $100 EACH**			
2	Annual payment	100		
3	r, interest rate	6%		
4				
5	**Year**	**Payment at end of year**	**Present value of payment**	
6	1	100	94.34	<-- =B6/(1+$B$3)^A6
7	2	100	89.00	<-- =B7/(1+$B$3)^A7
8	3	100	83.96	
9	4	100	79.21	
10	5	100	74.73	
11				
12	**Present value of all payments**			
13	Summing the present values		421.24	<-- =SUM(C6:C10)
14	Using Excel's PV function		421.24	<-- =PV(B3,5,-B2)
15	Using Excel's NPV function		421.24	<-- =NPV(B3,B6:B10)

5 future payments of $100 each, interest rate 6%.

Note three ways of getting the present value (cells C13:C15)

Net present value (NPV)

- Net present value of series of future cash flows is the pv of the cash flows minus the initial investment required to obtain them.
- Example: Pay $1,000 today to get $100 in year 1, $150 in year 2, …, $300 in year 5. Discount rate = 10%.

$$NPV = -1{,}000 + \frac{100}{1.10} + \frac{150}{(1.10)^2} + \frac{200}{(1.10)^3} + \frac{250}{(1.10)^4} + \frac{300}{(1.10)^5}$$

	A	B	C	D
1	**CALCULATING NET PRESENT VALUE (NPV) WITH EXCEL**			
2	r, interest rate	10%		
3				
4	**Year**	**Payment**	**Present value**	
5	0	-1,000	-1000.00	
6	1	100	90.91	<-- =B6/(1+$B$2)^A6
7	2	150	123.97	<-- =B7/(1+$B$2)^A7
8	3	200	150.26	
9	4	250	170.75	
10	5	300	186.28	
11				
12	**NPV**			
13	Summing the present values		-277.83	<-- =SUM(C5:C10)
14	Using Excel's NPV function		-277.83	<-- =B5+NPV($B$2,B6:B10)

The Net Present Value = -$777.83: The cost of $1,000 is $277.83 more than the Present Value of the future cash flows.

Therefore, it's not worth spending $1,000 to buy the future cash flows. The NPV < 0!

Using NPV to make a "Yes-No" investment decision

- An investment is worthwhile if its NPV>0.

	A	B	C
1	**MAKING THE "YES-NO" DECISION USING NPV**		
2	Discount rate	11%	
3			
4	Year	Cash flow	
5	0	-800	
6	1	200	
7	2	400	
8	3	300	
9	4	150	
10	5	900	
11			
12	NPV	557.10	<-- =B5+NPV(B2,B6:B10)

Investment is worthwhile since its NPV > 0

- Note that this depends both on the cash flows and the discount rate!

Using NPV to choose between investments

- When faced with two mutually-exclusive investments, choose the one with the largest NPV

	A	B	C	D
1	**USING NPV TO CHOOSE BETWEEN INVESTMENTS**			
2	Discount rate	15%		
3				
4	**Year**	**Investment A**	**Investment B**	
5	0	-800	-800	
6	1	250	600	
7	2	500	200	
8	3	200	100	
9	4	250	500	
10	5	300	300	
11				
12	NPV	219.06	373.75	<-- =NPV(B2,C6:C10)+C5

Investment B is preferred to Investment A

Internal rate of return (IRR)

- IRR is the discount rate for which the NPV = 0.
- Excel has an **IRR** function

	A	B	C	D	E
1	**COMPUTING IRR WITH EXCEL**				
2	**Year**	**Payment**			
3	0	-1,000			
4	1	100			
5	2	200			
6	3	300			
7	4	400			
8	5	500			
9					
10	IRR	12.01%	<-- =IRR(B3:B8)		
11					
12	**Discount rate**	**NPV**			
13	0.00%	500.00	<-- =NPV(A13,$B$4:$B$8)+$B$3		
14	1.00%	446.37	<-- =NPV(A14,$B$4:$B$8)+$B$3		
15	2.00%	395.37	<-- =NPV(A15,$B$4:$B$8)+$B$3		
16	3.00%	346.85			
17	4.00%	300.65			
18	5.00%	256.64			
19	6.00%	214.69			
20	12.01%	0.00			
21	8.00%	136.51			
22	9.00%	100.07			
23	10.00%	65.26			
24	11.00%	31.99			
25	12.00%	0.18			
26	14.00%	-59.38			
27	16.00%	-113.99			
28	18.00%	-164.16			
29	20.00%	-210.33			
30					
31					
32					
33					



Using IRR to make a “Yes-No” investment decision

- An investment is worthwhile if its IRR>discount rate.

	A	B	C
1	**MAKING THE "YES-NO" DECISION USING IRR**		
2	Discount rate	11%	
3			
4	Year	Cash flow	
5	0	-800	
6	1	200	
7	2	400	
8	3	300	
9	4	150	
10	5	900	
11			
12	IRR	31.19%	<-- =IRR(B5:B10)

Investment is worthwhile since its IRR>11% (the discount rate)

- Note that this depends both on the cash flows and the discount rate!

Using IRR to choose between investments

- When faced with two mutually-exclusive investments, choose the one with the largest IRR

	A	B	C	D
1	**USING IRR TO CHOOSE BETWEEN INVESTMENTS**			
2	**Year**	**Investment A cash flows**	**Investment B cash flows**	
3	0	-1,000.00	-1,000.00	
4	1	450.00	550.00	
5	2	425.00	300.00	
6	3	350.00	475.00	
7	4	450.00	200.00	
8				
9	IRR	24.74%	22.26%	<-- =IRR(C3:C7)

Investment A is preferred to Investment B

NPV and IRR

	"Yes or No": Choosing whether to undertake a single investment	"Investment ranking": Comparing two investments that are mutually exclusive
NPV criterion	The investment should be undertaken if its NPV > 0:	Investment A is preferred to investment B if NPV(A) > NPV(B)
IRR criterion	The investment should be undertaken if its IRR > r, where r is the appropriate discount rate.	Investment A is preferred to investment B if IRR(A) > IRR(B).

Chapter 4: IRR and NPV may not always rank two investments the same. In this case, use NPV to rank the investments.

Computing annual "flat" payments on a loan

- You borrow $10,000 for 5 years
- Interest rate 7%
- Bank wants same sum X repaid each year
- How to compute X?

$$10,000 = \underbrace{\frac{X}{1.07} + \frac{X}{(1.07)^2} + \frac{X}{(1.07)^3} + \frac{X}{(1.07)^4} + \frac{X}{(1.07)^5}}_{\text{Present value of repayments should equal initial sum borrowed}}$$

Use **PMT** to compute loan payments

	A	B	C
1	**PMT TO COMPUTE FLAT LOAN PAYMENTS**		
2	Interest rate	7%	
3	Loan term	5	<-- Years
4	Loan principal	10,000	
5	Annual flat payment	2,438.91	<-- =PMT(B2,B3,-B4)
6			
7	Present value of payments = Loan principal		
8	Year	Payment	
9	1	2,438.91	<-- =$B$5
10	2	2,438.91	<-- =$B$5
11	3	2,438.91	<-- =$B$5
12	4	2,438.91	
13	5	2,438.91	
14			
15	PV of payments	10,000.00	<-- =NPV(B2,B9:B13)

Loan amortization table

- The “flat” loan payment pays off the loan over the loan term:

	A	B	C	D	E	F
1	**PMT AND LOAN AMORTIZATION TABLE**					
2	Interest rate	7%				
3	Loan term	5	<-- Years			
4	Loan principal	10,000				
5	Annual flat payment	2,438.91	<-- =PMT(B2,B3,-B4)			
6						
7	**Loan Amortization Table**					
8	**Year**	**Principal, beginning of year**	**Payment, end of year**	**Part of payment that is interest**	**Part of payment that is principal**	
9	1	10,000.00	2,438.91	700.00	1,738.91	<-- =C9-D9
10	2	8,261.09	2,438.91	578.28	1,860.63	
11	3	6,400.46	2,438.91	448.03	1,990.87	
12	4	4,409.59	2,438.91	308.67	2,130.24	
13	5	2,279.35	2,438.91	159.55	2,279.35	
14						
15	Cell B10 contains formula			Cell D10 contains formula		
16	=B9-E9			=$B$2*B10		

Saving for the future

- Mario wants to buy a $20,000 car in 2 years.
- Plans to save X today and X in one year
- Interest rate = 8%



FV of X = cost of car

We show 3 ways to do this calculation

3 solutions to Mario's problem

- Trial and error
- **Goal seek**
- Excel's **PMT** function

Trial and error

	A	B	C	D	E
1	**HELPING MARIO SAVE FOR A CAR**				
2	Deposit, X	5,000.00			
3	Interest rate	8.00%			
4	**Year**	**In bank, before deposit**	**Deposit or withdrawal**	**Total at beginning of year**	**End of year with interest**
5	0	0.00	5,000.00	5,000.00	5,400.00
6	1	5,400.00	5,000.00	10,400.00	11,232.00
7	2	11,232.00	(20,000.00)	(8,768.00)	(9,469.44)
8					
9		NPV of all deposits and payments	-7,517.15	<-- =C5+NPV(B3,C6:C7)	

X = $5,000 is too little. The NPV < 0

	A	B	C	D	E
1	**HELPING MARIO SAVE FOR A CAR**				
2	Deposit, X	9,000.00			
3	Interest rate	8.00%			
4	**Year**	**In bank, before deposit**	**Deposit or withdrawal**	**Total at beginning of year**	**End of year with interest**
5	0	0.00	9,000.00	9,000.00	9,720.00
6	1	9,720.00	9,000.00	18,720.00	20,217.60
7	2	20,217.60	(20,000.00)	217.60	235.01
8					
9		NPV of all deposits and payments	186.56	<-- =C5+NPV(B3,C6:C7)	

X = $9,000 is too much. The NPV > 0

"Playing" with the numbers: When X = $8,903.13, the NPV (cell C9) = 0

Use **Goal Seek**

	A	B	C	D	E
1	**HELPING MARIO SAVE FOR A CAR**				
2	Deposit, X	9,000.00			
3	Interest rate	8.00%			
4	**Year**	**In bank, before deposit**	**Deposit or withdrawal**	**Total at beginning of year**	**End of year with interest**
5	0	0.00	9,000.00	9,000.00	9,720.00
6	1	9,720.00	9,000.00	18,720.00	20,217.60
7	2	20,217.60	(20,000.00)	217.60	235.01
8					
9		NPV of all deposits and payments	186.56	<-- =C5+NPV(B3,C6:C7)	
10					
11					
12					
13					
14					



Goal Seek will compute X = $8,903.13

Using **PMT** to compute X

	A	B	C
1	**HELPING MARIO SAVE FOR A CAR using Excel PMT function**		
2	Goal	20,000.00	<-- The cost of the car
3	When to reach the goal?	2	<-- The year in which Mario wants to buy the car
4	Interest rate	8.00%	
5	Deposit, X	8,903.13	<-- =PMT(B4,B3,,-B2,1)



Use **FV** box on **PMT** function to compute the beginning-of-period payment that gives $20,000 in two years.

Note that **Type** = 1, indicating beginning-of-period payments.

Saving for future: More complicated problems

- Linda is 10 years old. Her parents want to save for 4 years of college, starting at 18.
- Her parents want to deposit $X today and on birthdays 11, 12, … 17.
- On birthdays 18, 19, 20, 21, they plan to take out $20,000—the cost of college.

	A	B	C	D	E
	SAVING FOR COLLEGE				
2	Interest rate	8%			
3	Annual deposit	4,000.00			
4	Annual cost of college	20,000			
5					
6	**Birthday**	**In bank on birthday, before deposit/withdrawal**	**Deposit or withdrawal at beginning of year**	**Total**	**End of year with interest**
7	10	0.00	4,000.00	4,000.00	4,320.00
8	11	4,320.00	4,000.00	8,320.00	8,985.60
9	12	8,985.60	4,000.00	12,985.60	14,024.45
10	13	14,024.45	4,000.00	18,024.45	19,466.40
11	14	19,466.40	4,000.00	23,466.40	25,343.72
12	15	25,343.72	4,000.00	29,343.72	31,691.21
13	16	31,691.21	4,000.00	35,691.21	38,546.51
14	17	38,546.51	4,000.00	42,546.51	45,950.23
15	18	45,950.23	-20,000.00	25,950.23	28,026.25
16	19	28,026.25	-20,000.00	8,026.25	8,668.35
17	20	8,668.35	-20,000.00	-11,331.65	-12,238.18
18	21	-12,238.18	-20,000.00	-32,238.18	-34,817.24
19					
20		NPV of all payments	-13,826.40	<-- =C7+NPV(B2,C8:C18)	

$4,000 per year is too little: The NPV of all savings and payments < 0.

Like Mario's problem: Can solve this with trial-and-error, **Goal Seek**, or **PMT** and **PV**. See PFE book for details.

Solution
Annual deposit = $6,227.78

	A	B	C	D	E
1	**SAVING FOR COLLEGE**				
2	Interest rate	8%			
3	Annual deposit	6,227.78			
4	Annual cost of college	20,000			
5					
6	**Birthday**	**In bank on birthday, before deposit/withdrawal**	**Deposit or withdrawal at beginning of year**	**Total**	**End of year with interest**
7	10	0.00	6,227.78	6,227.78	6,726.00
8	11	6,726.00	6,227.78	12,953.77	13,990.08
9	12	13,990.08	6,227.78	20,217.85	21,835.28
10	13	21,835.28	6,227.78	28,063.06	30,308.10
11	14	30,308.10	6,227.78	36,535.88	39,458.75
12	15	39,458.75	6,227.78	45,686.52	49,341.45
13	16	49,341.45	6,227.78	55,569.22	60,014.76
14	17	60,014.76	6,227.78	66,242.54	71,541.94
15	18	71,541.94	-20,000.00	51,541.94	55,665.29
16	19	55,665.29	-20,000.00	35,665.29	38,518.52
17	20	38,518.52	-20,000.00	18,518.52	20,000.00
18	21	20,000.00	-20,000.00	0.00	0.00
19					
20		NPV of all payments	0.00	<-- =C7+NPV(B2,C8:C18)	